

9/6.7

Simulation de gestion d'un aéroport

L'Amstrad CPC, comme la plupart de ses concurrents micro-ordinateurs, possède une ludothèque très complète concernant des simulateurs de vol.

Mais, qui dit simulateur de vol, dit aussi décollage et atterrissage.

Or, sur quasiment tous les simulateurs, les gestions des procédures de décollage et d'atterrissage sont laissées à l'initiative complète du pilote ; alors que dans la réalité, le trafic aérien est régulé avec une extrême prudence par des contrôleurs aériens qualifiés, dans des tours de contrôle possédant toute l'électronique nécessaire à la fiabilité des manœuvres.

Nous avons ainsi décidé de combler cette lacune en vous proposant un programme qui vous placera dans la situation d'un contrôleur aérien, chargé de réguler le trafic d'un aéroport des environs de Paris.

MISE EN SITUATION ET BUT DU JEU

Vous voici donc aux « commandes » des ordinateurs sophistiqués de la tour de contrôle de l'aéroport Outinp-Sud.

Votre rôle consiste à permettre le décollage de quinze avions en attente et à libérer complètement les couloirs aériens entourant l'aéroport, en permettant les manœuvres d'atterrissage des différents avions qui se présentent sur l'aéroport.

L'écran de jeu

L'écran de jeu est divisé en quatre fenêtres qui permettront de visualiser :

- l'aéroport, les couloirs aériens, les pistes et les avions en attente ;
- l'état de l'aéroport qui permettra de connaître l'avancement du jeu et la météo ;
- l'état de chaque zone de l'aéroport ;
- le dialogue avec les ordinateurs.

L'aéroport

L'aéroport se compose de quatre zones aériennes (Nord, Est, Sud et Ouest), chacune d'elles étant divisée en quatre couloirs d'altitudes différents (1, 2, 3 et 4), comme les représente la figure 1.




Fig. 1

La rose des vents est rappelée en haut à gauche.

Dans chacune des zones et couloirs seront positionnés les différents avions au fur et à mesure des affectations des positions, par vos soins.

Lors de chaque décollage ou atterrissage, une petite animation viendra y agrémenter l'apparente staticité de la visualisation.

Dans cette fenêtre sont aussi visualisées les pistes, au nombre de quatre, dans la direction correspondante à chacun des points cardinaux.

Les pistes Est et Ouest sont réservées à l'atterrissage des avions ; la piste Nord, au décollage ; la piste Sud est par contre en réfection et est impraticable hélas pour toute la durée du jeu (voir les flèches de direction rappelant le sens de déplacement sur les pistes, en figure 1).

L'état de l'aéroport

Vous pourrez suivre dans cette fenêtre l'état d'avancement du jeu en y notant :

- le nombre d'avions en attente de décollage ;
- le nombre d'avions en attente dans les différents couloirs aériens ;
- le nombre d'avions dont vous avez assuré l'atterrissage.

Cette fenêtre indiquera aussi la direction que prend le vent pour le tour de jeu, ainsi que l'état de la piste.

- *Le vent*

Nous avons apporté dans ce jeu une importance non négligeable au vent et plusieurs possibilités pourront apparaître dans la direction de celui-ci : l'Est, l'Ouest, et le Sud.

Il arrivera parfois que la force du vent soit nulle. Par contre, notre aéroport sera particulièrement protégé d'un vent en direction du Nord, de part sa situation géographique.

L'importance du vent influencera, si vous êtes un contrôleur aérien consciencieux dans votre travail, vos choix de décollage ou atterrissage, car nous avons considéré que, pour réussir chacune de ces manœuvres sans coup férir, un avion doit se présenter face au vent.

Ainsi un atterrissage ne pourra être effectué sur la piste Ouest que si la direction prise par le vent est aussi l'Ouest. De même pour la piste Est, le vent doit aussi se diriger dans la direction Est.

Les décollages seront assurés sur la piste Nord si le vent par contre se dirige vers le Sud (rappelez-vous : face au vent).

Bien sûr, si le vent est nul, toute manœuvre de décollage ou d'atterrissage sera réalisée en toute sécurité (les pilotes sont supposés rodés à ces manœuvres).

- *L'état de la piste*

Pas de chance pour vos débuts dans le contrôle aérien, vous prenez votre fonction en plein milieu d'un hiver rigoureux et, malgré la qualité du personnel technique au sol, la piste est parfois gelée (voir figure 3), ce qui rend scabreuse, voir impossible toute manœuvre d'atterrissage ou de décollage.

Par contre, si la piste est réputée sèche, les manœuvres seront réalisées en toute sécurité (si tous les autres paramètres sont respectés).

Il sera bien sûr possible de passer outre les indications de ces deux paramètres, mais cela fera prendre de gros risques aux passagers

des avions concernés, les pilotes ne possédant pas toujours le sang-froid nécessaire aux manœuvres que vous indiquerez.

L'état des zones

Il vous sera possible de consulter sur cette fenêtre l'état de chacune des zones de l'aéroport afin de connaître les couloirs occupés et les autonomies en carburant disponibles pour chaque avion présent.

Le contenu de cette fenêtre sera développé plus en détail lors de l'étude de la commande correspondante.

La fenêtre de dialogue

Cette fenêtre, encore vierge sur la figure 1, se situe en bas à gauche de l'écran et vous permettra d'entrer vos ordres ou vos commandes pour permettre les manœuvres, ce que nous allons développer ci-dessous.

LE JEU

L'arrivée des premiers avions

Suite à l'écran de présentation, cinq avions se présentent aux abords de l'aéroport, et vous devez de suite leur trouver une place dans les différents couloirs aériens (voir figure 2).




Fig. 2

Lors de son arrivée, un avion déclare son type (onze sont en circulation, depuis le Boeing jusqu'au Tupolev, en passant par notre européen Airbus et autres Caravelle).

Il annonce aussi la quantité de carburant dont il dispose encore, suite à son voyage. Cette quantité, par multiple de cent (pour simplifier la gestion), pourra être comprise entre 200 et 900 litres de kérosène.

Bien sûr, la quantité de carburant sera déterminante sur le temps d'autonomie de l'avion en altitude, et moins il y en aura, plus il sera urgent d'autoriser l'atterrissage, en positionnant l'avion concerné dans un couloir relativement bas, et proche d'une piste potentiellement praticable.

L'ordinateur attend, pour chaque avion, dans l'ordre suivant :

- la zone de positionnement (Nord, Ouest, Est ou Sud) ;
- le couloir (1, 2, 3 ou 4).

L'acquisition de ces données est gérée dynamiquement, avec prise en compte immédiate de chacun des paramètres désignés.

Ainsi, si vous frappez sur 0 pour Ouest, l'ordinateur demande immédiatement l'altitude du couloir correspondant, car le pilote recevant l'ordre a déjà amorcé la manœuvre dans la direction choisie. Aucune correction n'est donc possible, cela étant dû à la vitesse des avions (vous savez peut-être que l'on considère comme une catastrophe aérienne le croisement de deux avions à quelques centaines de mètres l'un de l'autre ? ! Sinon, vous en subirez les conséquences).

La procédure d'acquisition est identique pour l'altitude du couloir aérien, hormis que l'altitude est codée par un chiffre entier compris entre 1 et 4.

Bien sûr, tout ordre erroné n'est pas pris en compte, comme le choix d'une altitude inexistante.

Lors de l'arrivée de ces avions, la vue radar de l'aéroport n'est pas régénérée au fur et à mesure, mais uniquement quand les indications de positionnement seront données aux cinq avions, comme indiqué sur la figure 3.

Vous veillerez aussi à ne pas positionner deux avions dans le même couloir aérien d'une même zone, sinon, c'est le télescopage assuré avec des conséquences en vies humaines.

Les ordres

Les cinq premiers avions étant positionnés, l'ordinateur se place en attente d'un ordre.

Les ordres, au nombre de cinq, sont les suivants :

- atterrissage d'un ou plusieurs avions ;
- mouvement d'un ou plusieurs avions ;
- décollage d'un avion ;
- état d'une zone ;
- pas d'action particulière.

Si la frappe d'une commande correspond à un ordre inexistant, l'ordinateur rappelle à l'écran la liste des ordres possibles, comme indiqué sur la figure 3.




Fig. 3

Les lettres à frapper pour lancer une commande sont signalées en vidéo inversée.

L'atterrissage

Il vous sera possible de commander l'atterrissage des avions par la commande A, signalée en vidéo inversée sur l'écran de rappel des ordres.

Afin de positionner correctement les avions, il vous est proposé le mouvement préliminaire d'un avion (voir figure 4).




Fig. 4

Ce mouvement consistera à déplacer un avion d'un couloir aérien à un autre pour le positionner face à la piste où l'atterrissage est possible, ou à écarter un avion gênant une manœuvre d'atterrissage.

Pour l'effectuer, il vous faudra entrer dans l'ordre :

- le nom de la zone de départ (E pour Est, N pour Nord...);
- l'altitude du couloir (1, 2, 3 ou 4) puis les coordonnées d'arrivée.

Si vous ne désirez pas effectuer ce mouvement, répondez F lors de l'entrée de la zone de départ.

Vous pourrez ensuite ordonner deux atterrissages au maximum en entrant les coordonnées de chacun des avions concernés.

Si vous désirez abandonner la manœuvre, il vous suffira d'entrer F lors de la demande de zone.

Bien sûr, certaines conditions sont requises pour réussir un atterrissage :

- le sens du vent ;
- l'état de la piste ;
- la présence ou non d'un avion sur un couloir aérien inférieur.

Si le sens du vent n'est pas en relation avec la zone d'atterrissage concernée, l'atterrissage est considéré risqué, comme indiqué, par exemple, sur la figure 4.

Si vous effectuez un atterrissage sur une piste gelée, il en sera de même, et il vous faudra compter sur le sang-froid du pilote de l'avion concerné.

Les risques que vous aurez pris, s'ils s'avèrent trop grands provoqueront irrémédiablement un accident et l'avion s'écrasera à l'atterrissage, avec son bilan en pertes humaines, comme c'est hélas prévisible dans ce cas de figure (voir figure 5).




Fig. 5

De même, lorsque vous ordonnez l'atterrissage d'un avion situé dans un couloir à une altitude supérieure ou égale à 2, il vous faudra vérifier que les couloirs inférieurs sont libres pour permettre le passage de l'avion, sous peine de provoquer une collision avec un autre avion en attente.

Enfin, si le décollage est réussi, une animation se déclenchera, présentant le décollage d'un avion sur le côté gauche de la fenêtre contenant la vue de l'aéroport.

L'état d'une zone

L'état d'une zone est accessible par l'appui sur la touche Z (voir figure 6).




Fig. 6

L'ordinateur vous demande alors la zone concernée, puis complète la fenêtre ETAT ZONE de droite.

Dans cette fenêtre, sont indiquées altitude par altitude, les quantités de carburant restant dans chaque avion en attente. Une zone libre étant repérée par une étoile.

Ainsi, sur la figure 6, sont présents en zone Est trois avions aux altitudes 1, 2 et 4, respectivement affectés de 100, 200 et 800 litres de kérosène ; les couloirs d'altitude 3 et 4 étant libres.

Cette fonction est importante pour connaître les capacités d'autonomie des avions et ainsi planifier les atterrissages.

Le décollage

Pour commander le décollage d'un avion, vous appuierez sur la touche D.

L'ordinateur vous demandera confirmation puis effectuera le décollage de l'avion en fonction des conditions météorologiques et de navigation :

- direction du vent ;
- état de la piste ;
- présence d'un avion dans un couloir supérieur.

Les décollages s'effectuent toujours sur la piste Nord, aussi, comme pour l'atterrissage, si le vent n'est pas en accord avec la piste concernée (c'est-à-dire en direction Sud), le décollage est risqué.

L'état de la piste est aussi important, et nous ne saurons trop vous conseiller d'effectuer les décollages sur piste sèche, plutôt que risquer la vie des passagers sur une piste gelée.

La figure 7 représente une de ces situations, qui s'est fort heureusement bien terminée.




Fig. 7

Il vous faudra aussi vérifier la présence d'un avion dans les couloirs aériens de la zone Nord, car si l'un de ceux-ci est occupé, le décollage se terminera irrémédiablement par une collision.

Une animation de décollage est aussi prévue sur la fenêtre supérieure en cas de réussite.

Les mouvements entre couloirs et zones

Afin de préparer vos avions pour les atterrissages, ou pour libérer la zone Nord pour un décollage, vous pourrez demander un mouvement d'avion par l'appui sur la touche **M**.

Il vous sera alors possible de déplacer jusqu'à trois avions.

L'ordinateur vous demande pour chacun d'eux les zones et altitudes de départ, ainsi que les coordonnées d'arrivée.




Fig. 8

Dans l'exemple de la figure 8, un premier avion situé en zone Est couloir 3 est déplacé en Ouest couloir 2.

De même en Est couloir 1, on déplace un avion en Ouest couloir 1.

Lors d'un mouvement dans une même zone, les conditions de libération des couloirs doivent être respectées. Ainsi, dans le deuxième mouvement, un avion était présent dans la zone Ouest couloir 1, ce qui a provoqué la collision signalée en figure 9, précédée d'une animation des deux avions se percutant.

```

Vous avez provoque une collision entre deux avions
en attente au dessus des pistes ....

Les secours ont denombre 177 victimes
Vous etes renvoye ...

Quelqu'un desire t'il prendre la place d'operateur
dans la tour de controle (O/N) ?

```

Fig. 9

Signalons que lors des mouvements, la fenêtre supérieure n'est régénérée qu'une fois tous les mouvements effectués, ce qui vous obligera à mémoriser les mouvements en cours, pour éviter les accidents.

Il est possible d'annuler la demande de déplacement en appuyant sur la touche F lors de la demande de zone de départ.

L'attente

Dans certains cas, la météo vous incitera à la prudence, et vous désirez peut-être passer votre tour en attendant une météo plus clémente, aussi, pas d'hésitation, appuyez sur la touche P pour attendre.

Bien sûr, l'attente soumet les avions aux règles de la consommation de carburant, et il faudra peut-être éviter celle-ci en provoquant un atterrissage en catastrophe.

Les réservoirs vides

A chaque tour de jeu, l'ordinateur remet à jour les capacités d'autonomie des avions en attente en imputant, à chacun d'eux, 100 litres de kérosène.

Dès que la capacité atteint la valeur zéro, inutile de compter faire planer l'avion pour l'atterrissage, celui-ci s'écrase irrémédiablement par une animation sur l'écran de jeu, comme représenté en figure 10.




Fig. 10

Ne pouvant plus contrôler l'avion, et à cause de la forte urbanisation entourant l'aéroport, la catastrophe a de fortes chances de se produire sur une ville environnante, avec son cortège de victimes, comme indiqué en figure 11.

```
L'AVION S'EST ECRASE SUR
                          NOGENT-SUR-MARNE

Cet accident a provoqué 144 victimes

Vous êtes renvoyé !!!!!

Quelqu'un veut-il prendre votre place (O/N)?
```

Fig. 11

Les arrivées intermédiaires

Entre chaque tour de jeu, au maximum trois avions pourront se présenter aux abords des pistes, aussi il vous faudra leur trouver une petite place, en fonction de leur capacité de carburant.

Souvenez-vous tout de même, pas plus d'un avion par couloir dans une même zone.

L'ALGORITHME DU PROGRAMME

Nous n'entrerons pas ici dans l'algorithme détaillé du programme, mais nous tenons à vous présenter le fonctionnement simplifié du programme en vous permettant d'étudier l'algorithme général ci-dessous.

- DEBUT
 - Initialiser les variables
 - Dessiner l'écran du jeu avec ses différentes fenêtres
 - POUR i DE 1 A 5
 - Proposer l'arrivée d'un avion
 - Attendre son positionnement
 - Vérifier s'il n'y a pas de collision
 - Mémoriser sa position
 - FINPOUR
 - TANTQUE la fin de la partie n'est pas atteinte
 - Régénérer l'écran de jeu
 - TANTQUE une commande valable n'est pas entrée

- Attendre l'entrée d'une commande
- FINTANTQUE
- SELONCAS
 - CAS « atterrissage »
 - PROCEDURE ATTERRISSAGE
 - CAS « décollage »
 - PROCEDURE DECOLLAGE
 - CAS « mouvement »
 - PROCEDURE MOUVEMENT
 - CAS « zone »
 - PROCEDURE ETAT ZONE
 - CAS « attendre »
 - PROCEDURE ATTENTE
- FINCAS
- Diminuer de 100 toutes les quantités de carburant des avions en attente
- Si un avion se trouve « à sec »
- ALORS
 - Annoncer la fin de la partie
- FINSI
- Si de nouveaux avions arrivent
- ALORS
 - POUR i ALLANT de 1 AU nombre d'avions
 - Signaler l'arrivée d'un avion
 - Acquérir la position d'attente
 - Vérifier s'il n'y pas collision
 - Positionner l'avion
 - FINPOUR
- FINSI
- FINTANTQUE
- FIN

Cet algorithme fait appel à cinq procédures insérées dans la structure de cas traitant le choix de la commande effectuée, aussi nous vous proposons de détailler trois d'entre elles, concernant les mouvements d'avions.

La procédure d'atterrissage

- PROCEDURE ATERRISSAGE
- PROCEDURE MOUVEMENT D'UN AVION
- POUR i DE 1 A 2
 - Acquérir les coordonnées d'un avion à atterrir
 - SI il y a un avion dans un couloir inférieur
 - Signaler la fin de la partie
 - CO
collision
 - FINCO
 - FINSI
 - SI le vent n'est pas dans la bonne direction
 - ALORS
 - SI la chance n'est pas avec vous
 - Signaler la fin de la partie
 - CO
l'avion s'écrase
 - FINCO
 - FINSI
 - FINSI
 - SI la piste est gelée
 - ALORS
 - SI la chance n'est pas avec vous
 - Signaler la fin de la partie
 - CO
l'avion s'écrase
 - FINCO
 - FINSI
 - FINSI
- FINPOUR
- SI il ne reste plus d'avion au décollage
 - ALORS
 - SI il ne reste plus d'avions en attente
 - ALORS
 - Signaler la fin de la partie
 - CO
la partie est gagnée
 - FINCO
 - FINSI
 - FINSI
- FINPROCEDURE

La procédure de décollage

- PROCEDURE DECOLLAGE
 - Demander confirmation du décollage
 - SI il y a confirmation
 - ALORS
 - SI il y a un avion dans un couloir aérien de la zone Nord
 - ALORS
 - Signaler la fin de la partie
 - CO
collision
 - FINCO
 - Animer le décollage
 - FINSI
 - FINPROCEDURE

La procédure de déplacement

- PROCEDURE MOUVEMENT
 - POUR i DE 1 A 3
 - Acquérir les coordonnées de départ
 - Acquérir les coordonnées d'arrivée
 - SI il y a un avion dans la case de destination
 - ALORS
 - Signaler la fin de la partie
 - CO
collision
 - FINCO
 - FINSI
 - SI le déplacement s'effectue dans la même zone
 - ALORS
 - SI il y a un avion dans un couloir traversé
 - ALORS
 - Signaler la fin de la partie
 - CO
collision
 - FINCO
 - FINSI
 - FINSI
 - FINPOUR
 - FINPROCEDURE

LE PROGRAMME

La plupart des jeux de type simulation sur micro-ordinateur requièrent un programme conséquent, et le nôtre n'échappe pas à la règle, aussi, nous ne saurons trop vous recommander de vous concentrer lors de la frappe du listing que nous vous proposons.

Les noms de variables devront être respectés, en portant un intérêt particulier à leur orthographe et leur type (ainsi, par souci d'économie de mémoire, les variables entières ont été largement utilisées, prenez donc soin de ne pas oublier le symbole % les affectant).

De même, n'hésitez pas à sauvegarder régulièrement les lignes entrées, et à arrêter la frappe lorsque l'énervement vous taquine les doigts, le programme est numéroté de 10 en 10 pour vous faciliter la frappe et la reprise ultérieure.

Dernier conseil, entrez toutes les remarques (REM) car celles-ci vous permettront, premièrement, de comprendre le programme pour éventuellement y effectuer des modifications ultérieures, deuxièmement, pour repérer une erreur de frappe lors de l'exécution.

```

10 REM *****
20 REM ** **
30 REM **      ICI PAPA TANGO      **
40 REM ** SIMULATION DE GESTION **
50 REM ** DE TOUR DE CONTROLE **
60 REM ** **
70 REM *****
80 REM
90 MODE 2
100 RANDOMIZE TIME
110 CALL &BB4E
120 REM
130 DIM attente%(4,4): REM les zones d'a
ttente
140 REM      : ;
150 REM      i -> altitude
160 REM      ---> direction
170 REM      1 = Nord
180 REM      2 = Ouest
190 REM      3 = Sud
200 REM      4 = Est
210 REM attente%(x,y): REM les zones d'a
ttente
220 REM
230 REM ** DEFINITION DESSINS AVIONS **
240 REM
250 SYMBOL AFTER 250
260 SYMBOL 250,0,0,0,0,63,255,0
270 SYMBOL 251,0,2,6,14,254,254,0,0
280 formeavion1$ = CHR$(250)+CHR$(251)
290 SYMBOL 252,0,64,96,112,127,127,0,0
300 SYMBOL 253,0,0,0,0,252,255,0,0
310 formeavion2$ = CHR$(252)+CHR$(253)
320 REM
330 REM ** LE CIEL EST VIDE **
340 REM
350 FOR i% = 0 TO 4
360   FOR j% = 0 TO 4
370     attente%(i%,j%) = -1
380   NEXT j%
390 NEXT i%
400 REM ** 15 AVIONS A DECOLLER **
410 nombredecollage% = 15
420 REM ** ARRIVEE 2 AVIONS MAXI **
430 arrivemaxi% = 2 : REM nombre maximum

```

```

d'avions qu'il puisse arriver
440 REM ** 1 AVION MAX A BOUGER **
450 bougemaxi% = 1 : REM nombre maximum
d'avions a bouger avant atterissage
460 REM ** 2 AVION MAX A ATTERIR **
470 atterimaxi% = 2 : REM nombre maximum
d'avions qui peuvent atterir
480 REM
490 REM ** INITIALISATION FENETRES **
500 REM
510 REM fenetre d'ordre
520 WINDOW #0,1,40,17,25
530 REM fenetre pistes et zones
540 WINDOW #1,1,80,1,15
550 REM fenetre contenu des zones
560 WINDOW #2,61,80,17,25
570 REM fenetre etat ciel et temps
580 WINDOW #3,41,59,17,25
590 REM
600 REM affichage fenetre 2
610 REM
620 PRINT #2,CHR$(24)
630 CLS #2
640 PRINT #2," ETAT ZONE "
650 PRINT #2,"-----";
660 PRINT #3,CHR$(24)
670 CLS #3
680 REM
690 REM affichage fenetre 3
700 REM
710 PRINT #3," ETAT AERDPORT"
720 PRINT #3,"-----";
730 PRINT #3,"Decollage:"
740 PRINT #3,"Attente :"
750 PRINT #3,"Atterris :"
760 PRINT #3
770 PRINT #3,"Vent      :"
780 PRINT #3,"Piste      :"
790 REM
800 REM *****
810 REM **
820 REM ** DESSIN DES PISTES **
830 REM ** ET DES ZONES D'ATTENTE **
840 REM **
850 REM *****

```

```
860 REM
870 WINDOW #1,1,65,1,15
880 CLS #1
890 REM
900 REM ** DESSIN PISTES **
910 REM
920 PLOT 300,296
930 DRAW 300,316
940 DRAW 280,316
950 DRAW 280,296
960 DRAW 250,296
970 DRAW 250,276
980 DRAW 280,276
990 DRAW 280,256
1000 DRAW 300,256
1010 DRAW 300,276
1020 DRAW 340,276
1030 DRAW 340,296
1040 DRAW 300,296
1050 TAG #1
1060 MOVE 260,293
1070 PRINT #1,CHR$(243);
1080 MOVE 320,293
1090 PRINT #1,CHR$(242);
1100 MOVE 286,314
1110 PRINT #1,CHR$(240);
1120 TAGOFF #1
1130 REM
1140 REM ** NUMEROTATION DES ZONES **
1150 REM
1160 FOR I% = 4 TO 1 STEP -1
1170 LOCATE #1,34,6-I%
1180 PRINT #1,I%
1190 NEXT I%
1200 FOR I% = 4 TO 1 STEP -1
1210 LOCATE #1,30 - 4 * I%,9
1220 PRINT #1,I%
1230 NEXT I%
1240 FOR I% = 1 TO 4
1250 LOCATE #1,43 + 4 * I%,6
1260 PRINT #1,I%
1270 NEXT I%
1280 FOR I% = 1 TO 4
1290 LOCATE #1,38,9 + I%
1300 PRINT #1,I%
```



```
1310 NEXT I%
1320 REM
1330 REM ** DESSIN DES ZONES D'ATTENTE *
*
1340 REM
1350 MOVE 300,296
1360 DRAW 500,395
1370 DRAW 500,177
1380 DRAW 300,276
1390 PLOT 500,177
1400 DRAW 100,177
1410 DRAW 280,276
1420 PLOT 100,177
1430 DRAW 100,395
1440 DRAW 280,296
1450 PLOT 100,395
1460 DRAW 500,395
1470 REM
1480 REM ** DESSIN ROSE DES VENTS **
1490 REM
1500 LOCATE #1,4,1
1510 PRINT #1,"N"
1520 LOCATE #1,4,2
1530 PRINT #1,CHR$(240)
1540 PRINT #1,"O ";CHR$(242); " ";CHR$(24
3); " E"
1550 LOCATE #1,4,4
1560 PRINT #1,CHR$(241)
1570 LOCATE #1,4,5
1580 PRINT #1,"S"
1590 REM
1600 REM ** ANIMATION **
1610 REM
1620 LOCATE #1,1,15
1630 FOR I% = 1 TO 60
1640     PRINT #1,SPC(1);formeavion2$;CHR
$(8);CHR$(8);
1650 NEXT I%
1660 PRINT #1,SPC(2);
1670 REM
1680 REM *****
1690 REM ** **
1700 REM ** ARRIVEE DES PREMIERS **
```



```
2140 GOSUB 8220 : REM determiner la mete
a
2150 GOSUB 7750 : REM complete l'ecran e
tat du ciel
2160 PRINT
2170 PRINT "Technicien operateur, a vos
ordres"
2180 REM
2190 REM
2200 REM attente ordre
2210 REM
2220 a$ = ""
2230 WHILE a$ = ""
2240     a$ = INKEY$
2250 WEND
2260 a$ = UPPER$(a$)
2270 REM
2280 REM traitement ordre valide
2290 REM
2300 flag% = 0
2310 IF a$ = "A" THEN GOSUB 2770 : REM A
tterissage demande
2320 IF a$ = "M" THEN GOSUB 3870 : REM M
ouvement d'avion demande
2330 IF a$ = "P" THEN GOSUB 4590 : REM P
as d'action - Attente
2340 IF a$ = "Z" THEN GOSUB 5140 : REM D
emande d'etat de zone
2350 IF a$ = "D" THEN GOSUB 5470 : REM D
ecollage avion
2360 IF flag% = 1 THEN GOSUB 7990 : REM
mise a jour avions
2370 IF flag% = 1 THEN GOSUB 4740 : REM
voir si nouveaux avions
2380 IF flag% = 1 THEN GOTO 2120
2390 IF flag% = 2 THEN GOTO 2160
2400 REM
2410 REM ****|*****|*****|*****
2420 REM **                               **
2430 REM ** RAPPEL DES ORDRES **
2440 REM **                               **
2450 REM *****
2460 REM
2470 PRINT
```

```
2480 PRINT "Utilisez les ordres suivants
commandant: "
2490 FOR pause% = 1 TO 200
2500 NEXT pause%
2510 PRINT CHR$(24); "A"; CHR$(24);
2520 PRINT "atterrissage d'un avion";
2530 PRINT SPC(18)
2540 PRINT CHR$(24); "M"; CHR$(24);
2550 PRINT "mouvement d'avions";
2560 PRINT SPC(22)
2570 PRINT CHR$(24); "D"; CHR$(24);
2580 PRINT "decollage d'un avion";
2590 PRINT SPC(20)
2600 PRINT "Etat d'une ";
2610 PRINT CHR$(24); "Z"; CHR$(24); "one";
2620 PRINT SPC(25)
2630 PRINT CHR$(24); "P"; CHR$(24);
2640 PRINT "pas d'action - attente";
2650 PRINT SPC(18)
2660 FOR pause% = 1 TO 100
2670 NEXT pause%
2680 REM
2690 GOTO 2160
2700 REM
2710 REM *****
2720 REM ** **
2730 REM ** DEMANDE D'ATERRISSAGE **
2740 REM ** **
2750 REM *****
2760 REM
2770 flag% = 1
2780 PRINT
2790 PRINT "Mouvement preliminaire (F si
non)
2800 nombremaximal% = bougemaxi%
2810 GOSUB 3910
2820 FOR nombre% = 1 TO atterimaxi%
2830 PRINT
2840 PRINT "Atterissage de l'avion si
tue"
2850 PRINT "Zone(O,E,Fin)?";
2860 REM
2870 a$ = ""
2880 WHILE a$ <> "O" AND a$ <> "E" AN
```

```
D a$ <> "F"
2890     a$ = ""
2900     WHILE a$ = ""
2910         a$ = INKEY$
2920     WEND
2930     a$ = UPPER$(a$)
2940 WEND
2950 REM
2960 IF a$ = "O" THEN position% = 2 :
PRINT "Ouest";
2970 IF a$ = "E" THEN position% = 4 :
PRINT "Est";
2980 IF a$ = "F" THEN nombre% = atter
imaxi% : GOTO 3580
2990 REM
3000 PRINT " Altitude?";
3010 altitude% = 0
3020 WHILE altitude% < 1 OR altitude%
> 4
3030     altitude% = VAL(INKEY$)
3040 WEND
3050 PRINT altitude%
3060 IF directionvent% = 1 THEN GOTO
3090
3070 IF position% <> directionvent% T
HEN GOTO 3690
3080 REM
3090 IF attente%(position%,altitude%)
= -1 THEN GOTO 3630
3100 REM
3110 REM ** voir avion dessous **
3120 REM
3130 FOR k% = altitude% TO 2 STEP -1
3140     IF attente%(position%,k% - 1)
<> -1 THEN GOTO 6840
3150 NEXT k%
3160 REM
3170 REM ** ATTERISSAGE AUTORISE **
3180 REM
3190 attente%(position%,altitude%) =
-1
3200 IF etatpiste% = 1 THEN GOTO 3490
3210 FOR i% = 7 TO 12
3220     FOR j% = 1 TO 9
```

```
3230         LOCATE #1,j%,i%
3240         PRINT #1,formeavion2$
3250         FOR k% = 1 TO 10
3260         NEXT k%
3270         LOCATE #1,j%,i%
3280         PRINT #1,SPC(2)
3290     NEXT j%:i%=i%+1
3300     i%=i%+1
3310     FOR j% = 8 TO 1 STEP -1
3320         LOCATE #1,j%,i%
3330         PRINT #1,formeavion1$
3340         FOR k% = 1 TO 10
3350         NEXT k%
3360         LOCATE #1,j%,i%
3370         PRINT #1,SPC(2)
3380     NEXT j%
3390 NEXT i%
3400 PRINT "Atterissage correct"
3410 nombreavion% = nombreavion% - 1
3420 nombreatteris% = nombreatteris%
+ 1
3430 IF nombreavion% = 0 AND nombrede
collage% = 0 THEN GOTO 8320 : REM gagne,
plus d'avion
3440 REM
3450 GOTO 3580:REM au suivant
3460 REM
3470 REM ** PISTE GELEE **
3480 REM
3490 PRINT
3500 PRINT "Atterissage en catastroph
e ... "
3510 FOR k% = 1 TO 3000
3520 NEXT k%
3530 PRINT CHR$(10)
3540 local% = INT(RND * 2)
3550 IF local% = 0 THEN GOTO 8550: av
ion ecrase
3560 PRINT "Ouf ! Atterissage reussi"
3570 REM
3580 NEXT nombre%
3590 RETURN
3600 REM
```



```
4010     positiondepart$ = UPPER$(positiondepart$)
4020     WEND
4030     IF positiondepart$ = "F" THEN RETURN
4040     PRINT positiondepart$; " , ";
4050     altitudedepart% = 0
4060     WHILE altitudedepart% < 1 OR altitudedepart% > 4
4070         altitudedepart% = VAL(INKEY$)
4080     WEND
4090     PRINT altitudedepart%
4100     IF positiondepart$ = "F" THEN nombre% = 3: GOTO 4450
4110     REM
4120     PRINT "Arrivee : ";
4130     position$ = ""
4140     altitude% = 0
4150     WHILE position$ <> "N" AND position$ <> "O" AND position$ <> "E" AND position$ <> "S"
4160         position$ = INKEY$
4170         position$ = UPPER$(position$)
4180     WEND
4190     PRINT position$; " , ";
4200     altitude% = 0
4210     WHILE altitude% < 1 OR altitude% > 4
4220         altitude% = VAL(INKEY$)
4230     WEND
4240     PRINT altitude%
4250     REM
4260     IF positiondepart$ = "N" THEN positiondepart% = 1
4270     IF positiondepart$ = "O" THEN positiondepart% = 2
4280     IF positiondepart$ = "S" THEN positiondepart% = 3
4290     IF positiondepart$ = "E" THEN positiondepart% = 4
4300     IF position$ = "N" THEN position% = 1
4310     IF position$ = "O" THEN position% = 2
4320     IF position$ = "S" THEN position%
```



```

% = 3
4330 IF position$ = "E" THEN position
% = 4
4340 REM
4350 REM ** VERIFICATIONS **
4360 REM
4370 IF attente%(positiondepart%, alti
tudedepart%) = -1 THEN GOTO 4500
4380 IF attente%(position%, altitude%)
<> -1 THEN GOTO 6840
4390 IF positiondepart% <> position%
THEN GOTO 4430
4400 FOR descente% = altitudedepart%
+ SGN(altitude% - altitudedepart%) TO al
titude% STEP SGN(altitude% - altitudedep
art%)
4410 IF attente%(positiondepart%, d
escente%) <> -1 THEN GOTO 6840
4420 NEXT descente%
4430 attente%(position%, altitude%) =
attente%(positiondepart%, altitudedepart%
)
4440 attente%(positiondepart%, altitud
edepart%) = -1
4450 NEXT nombre%:REM au suivant
4460 REM
4470 REM ** ERREUR **
4480 REM
4490 RETURN
4500 PRINT "Pas d'avion a cette position
"
4510 GOTO 4130
4520 REM
4530 REM ****
4540 REM ** **
4550 REM ** ATTENTE DU DEGEL **
4560 REM ** **
4570 REM ****
4580 REM
4590 flag% = 1
4600 PRINT
4610 PRINT "Attente ...."
4620 FOR i% = 1 TO 3000
4630 NEXT i%

```

```

4640 PRINT
4650 REM
4660 RETURN
4670 REM *****
4680 REM **
4690 REM ** VDIR SI DE NOUVEAUX **
4700 REM ** AVIONS ARRIVENT **
4710 REM **
4720 REM *****
4730 REM
4740 local% = INT (RND * 2)
4750 IF local% = 0 THEN RETURN:REM pas d
'avion a l'arrivee
4760 local% = INT (RND * arrivemaxi%) +
1
4770 REM
4780 FOR arrive% = 1 TO local%
4790 PRINT
4800 GOSUB 6220:REM recherche carbura
nt
4810 REM
4820 PRINT "Demande atterissage pour
";
4830 PRINT typeavion$
4840 PRINT "Niveau kerosene: ";
4850 PRINT carburant% * 100;" litres";
4860 REM
4870 PRINT "Zone (N,D,S,E) ";
4880 a$ = ""
4890 WHILE a$ <> "N" AND a$ <> "O" AN
D a$ <> "S" AND a$ <> "E"
4900 a$ = INKEY$
4910 a$ = UPPER$(a$)
4920 WEND
4930 PRINT a$;
4940 position$ = a$
4950 PRINT " Altitude (1 a 4) ";
4960 a% = 0
4970 WHILE a% < 1 OR a% > 4
4980 a$ = INKEY$
4990 a% = VAL(a$)
5000 WEND
5010 PRINT a%
5020 altitude% = a%

```

```

5030     GOSUB 6750 : REM voir si positio
n valide
5040 NEXT arrive%
5050 REM
5060 RETURN
5070 REM
5080 REM ****|**|**|**|**|**|**|**|**|**|**
5090 REM **                                     **
5100 REM **          ETAT ZONE                **
5110 REM **                                     **
5120 REM ****|**|**|**|**|**|**|**|**|**|**
5130 REM
5140 CLS #2
5150 flag% = 2
5160 PRINT #2,"          ETAT ZONE "
5170 PRINT #2,"-----";
5180 PRINT
5190 PRINT "Etat de la zone (N/O/S/E)?"
5200 a$ = ""
5210 WHILE a$ <> "N" AND a$ <> "O" AND a
$ <> "S" AND a$ <> "E"
5220     a$ = INKEY$
5230     a$ = UPPER$(a$)
5240 WEND
5250 PRINT #2,SFC(8);
5260 IF a$ = "N" THEN PRINT #2,"NORD":a%
= 1
5270 IF a$ = "O" THEN PRINT #2,"OUEST":a
% = 2
5280 IF a$ = "S" THEN PRINT #2,"SUD":a%
= 3
5290 IF a$ = "E" THEN PRINT #2,"EST":a%
= 4
5300 PRINT #2
5310 FOR i% = 4 TO 1 STEP -1
5320     PRINT #2,i%;
5330     PRINT #2," -> ";
5340     IF attente%(a%,i%) = -1 THEN PRI
NT #2, " *":GOTO 5360
5350     PRINT #2,attente%(a%,i%) * 100;"
litres"
5360 NEXT i%
5370 REM
5380 RETURN
5390 REM

```

```
5400 REM **************
5410 REM **                **
5420 REM ** DECOLLAGE D'UN **
5430 REM ** AVION SUR LA  **
5440 REM ** PISTE NORD    **
5450 REM **                **
5460 REM *****
5470 flag% = 1
5480 PRINT
5490 PRINT "Demande de confirmation de"
5500 PRINT "Decollage (O/N) ";
5510 a$ = ""
5520 WHILE a$ <> "O" AND a$ <> "N"
5530     a$ = INKEY$
5540     a$ = UPPER$(a$)
5550 WEND
5560 IF a$ = "O" THEN GOTO 5600
5570 PRINT
5580 PRINT "*** ANNULATION ***"
5590 RETURN
5600 IF directionvent% = 3 OR directionvent% = 1 THEN GOTO 5750
5610 PRINT
5620 PRINT "Vent en direction ";
5630 PRINT vent$
5640 PRINT "Decollage risque ..."
5650 FOR i% = 1 TO 3000
5660 NEXT i%
5670 local% = INT(RND * 3) + 1
5680 IF local% = 2 THEN GOTO 5730
5690 PRINT "Une rafale rabat l'avion ..."
"
5700 FOR i% = 1 TO 4000
5710 NEXT i%
5720 GOTO 8550
5730 PRINT "Ouf ! Decollage reussi"
5740 GOTO 5900
5750 GOSUB 6220
5760 PRINT typeavion$;" : ";
5770 PRINT "Ok, Alpha-Tango"
5780 IF etatpiste% = 0 THEN GOTO 5900
5790 PRINT "Cabine gelee ... essai decollage manuel"
5800 local% = INT(RND * 4)
5810 IF local% = 1 THEN GOTO 5900
```

```
5820 PRINT "Impossible atteindre vitesse
optimale"
5830 PRINT "Fin de piste approche ...."
5840 FOR i% = 1 TO 8000
5850 NEXT i%
5860 GOTO 8550
5870 REM
5880 REM ** verification zone libre **
5890 REM
5900 FOR i% = 1 TO 4
5910     IF attente%(1,i%) <> -1 THEN GOT
O 6840
5920 NEXT i%
5930 FOR i% = 12 TO 7 STEP -1
5940     FOR j% = 1 TO 9
5950         LOCATE #1,j%,i%
5960         PRINT #1,formeavion2$
5970         FOR k% = 1 TO 10 : NEXT k%
5980         LOCATE #1,j%,i%
5990         PRINT #1,SPC(2)
6000     NEXT j%
6010     i% = i% - 1
6020     FOR j% = 8 TO 1 STEP -1
6030         LOCATE #1,j%,i%
6040         PRINT #1,formeavion1$
6050         FOR k% = 1 TO 10 : NEXT k%
6060         LOCATE #1,j%,i%
6070         PRINT #1,SPC(2)
6080     NEXT j%
6090 NEXT i%
6100 nombredecollage% = nombredecollage%
- 1
6110 IF nombreavion% = 0 AND nombredecol
lage% = 0 THEN GOTO 8320 : REM gagne, pl
us d'avions
6120 RETURN
6130 REM
6140 REM ****
6150 REM **
6160 REM ** RECHERCHE TYPE ET **
6170 REM ** CARBURANT D'UN AVION **
6180 REM ** A L'ARRIVEE **
6190 REM **
6200 REM *****
6210 REM
```

```
6220 local% = INT(RND * 12)
6230 typeavion$ = "DC-10"
6240 IF local% = 1 THEN typeavion$ = "74
7"
6250 IF local% = 2 THEN typeavion$ = "73
7"
6260 IF local% = 3 THEN typeavion$ = "A-
310"
6270 IF local% = 4 THEN typeavion$ = "A-
320"
6280 IF local% = 5 THEN typeavion$ = "AT
R-42"
6290 IF local% = 6 THEN typeavion$ = "TU
POLEV 114"
6300 IF local% = 7 THEN typeavion$ = "FO
KKER F-28"
6310 IF local% = 8 THEN typeavion$ = "CO
NCORDE"
6320 IF local% = 9 THEN typeavion$ = "FA
LCON 50"
6330 IF local% = 10 THEN typeavion$ = "C
ARAVELLE 3"
6340 IF local% = 11 THEN typeavion$ = "G
ALAXY C-5A"
6350 REM
6360 REM carburant
6370 REM
6380 carburant% = INT(RND * 8) + 2
6390 REM
6400 RETURN
6410 REM
6420 REM *****
6430 REM ** **
6440 REM ** RECHERCHE DU SENS **
6450 REM ** DU VENT **
6460 REM ** **
6470 REM *****
6480 REM
6490 directionvent% = INT(RND * 4) + 1
6500 IF directionvent% = 1 THEN vent$ =
"Nu1"
6510 IF directionvent% = 2 THEN vent$ =
"Ouest"
6520 IF directionvent% = 3 THEN vent$ =
```



```
6850 FOR i% = 1 TO 20
6860     PRINT #1,SPC(i%);formeavion2$;
6870     PRINT #1,SPC(40 - 2 * i%);formeav
vion1$;SPC(2)
6880     PRINT #1,CHR$(13);
6890 NEXT i%
6900 FOR i% = 1 TO 10
6910     FOR j% = 1 TO 50
6920         INK 0,26
6930     NEXT j%
6940     FOR j% = 1 TO 50
6950         INK 0,0
6960     NEXT j%
6970 NEXT i%
6980 MODE 0
6990 PRINT:PRINT:PRINT
7000 PRINT "COLLISION"
7010 FOR i% = 1 TO 5000
7020 NEXT i%
7030 MODE 2
7040 PRINT:PRINT
7050 PRINT "Vous avez provoque une colli
sion entre deux avions"
7060 PRINT "en attente au dessus des pis
tes ...."
7070 PRINT
7080 PRINT
7090 PRINT "Les secours ont denombre ";
7100 nombrevictime% = INT(150 * RND) + 1
50
7110 PRINT CHR$(24);nombrevictime%;
7120 PRINT " victimes ";CHR$(24)
7130 PRINT
7140 PRINT "Vous etes renvoye ..."
7150 FOR i% = 1 TO 8
7160     PRINT
7170 NEXT i%
7180 PRINT "Quelqu'un desire t'il prendr
e la place d'operateur"
7190 PRINT "dans la tour de controle (O/
N) ?"
7200 a$ = ""
7210 WHILE a$ = ""
7220     a$ = INKEY$
7230 WEND
```



```

B080 FOR I% = 1 TO 4
B090   FOR J% = 1 TO 4
B100     IF attente%(i%,j%) = 0 THEN G
OTO 9110
B110   NEXT J%
B120 NEXT I%
B130 REM
B140 RETURN
B150 REM
B160 REM *****
B170 REM **                               **
B180 REM **   ETUDE DE LA METEO   **
B190 REM **                               **
B200 REM *****
B210 REM
B220 GOSUB 6490 : REM recherche du sens
du vent
B230 GOSUB 6630 : REM recherche de l'eta
t de la piste
B240 REM
B250 RETURN
B260 REM *****
B270 REM **                               **
B280 REM **   PLUS D'AVION   **
B290 REM **                               **
B300 REM *****
B310 REM
B320 CLS
B330 PRINT "          BRAVO"
B340 PRINT
B350 PRINT "Tous les avions en attente"
B360 PRINT "ont atterri, le patron vous"
B370 PRINT "attend pour vous feliciter e
t"
B380 PRINT "vous augmenter ..."
B390 PRINT "A moins que vous ne preferie
z"
B400 PRINT "retenter votre chance (D/N)
?"
B410 a$ = ""
B420 WHILE a$ = ""
B430   a$ = INKEY$
B440 WEND
B450 a$ = UPPER$(a$)

```



```
9280     LOCATE i%,i%
9290     PRINT formeavion2$
9300     FOR k% = 1 TO 30
9310     NEXT k%
9320     LOCATE i%,i%
9330     PRINT SPC(2)
9340 NEXT i%
9350 FOR I% = 1 TO 30
9360     INK 0,26
9370     INK 1,0
9380     FOR J% = 1 TO 30
9390     NEXT J%
9400     INK 0,0
9410     INK 1,26
9420     FOR J% = 1 TO 30
9430     NEXT J%
9440 NEXT I%
9450 MODE 2
9460 LOCATE 10,5
9470 PRINT "L'AVION S'EST ECRASE SUR "
9480 LOCATE 25,7
9490 I% = INT(RND * 8) + 1
9500 IF I% = 1 THEN PRINT "LA TOUR DE CO
NTROLE"
9510 IF I% = 2 THEN PRINT "CERGY-PONTOIS
E"
9520 IF I% = 3 THEN PRINT "NOGENT-SUR-MA
RNE"
9530 IF I% = 4 THEN PRINT "NEUILLY"
9540 IF I% = 5 THEN PRINT "EVRY"
9550 IF I% = 6 THEN PRINT "NANTERRE"
9560 IF I% = 7 THEN PRINT "VERSAILLES"
9570 IF I% = 8 THEN PRINT "CORBEIL/ESSON
NE"
9580 PRINT
9590 PRINT
9600 PRINT "Cet accident a provoqe ";
9610 i% = INT(RND * 1000) + 800
9620 PRINT CHR$(24);i%;" ";
9630 PRINT "victimes ";CHR$(24)
9640 PRINT:PRINT
9650 PRINT "Vous etes renvoye ！！！！"
9660 PRINT:PRINT
9670 PRINT "Quelqu'un veut-il prendre vo
tre place (O/N)?"
```

```
9680 a$=""
9690 WHILE a$ = ""
9700     a$ = INKEY$
9710 WEND
9720 a$ = UPPER$(a$)
9730 IF a$ = "0" THEN RUN
9740 PRINT:PRINT
9750 STOP
```

ETUDE DU PROGRAMME

Nous allons étudier l'allure générale du programme, afin de repérer les principaux traitements effectués et les variables utilisées, ce qui vous permettra d'effectuer quelques modifications pour faire évoluer le jeu.

Etude du listing

Lignes 10 à 310 : initialisation des variables, tableaux et dessins des avions. Le tableau contenant les avions en attente au-dessus des pistes est appelé ici `attente%(i, j)`.

Lignes 350 à 390 : toutes les cases du tableau sont initialisées à la valeur -1 pour signaler qu'il n'y a pas d'avion. Pourquoi pas zéro ? Réponse : nous avons décidé de placer dans ce tableau des quantités en kérosène des avions ; aussi, la valeur zéro correspondra-t-elle à un avion en perdition, et non à une position vide.

Lignes 410 à 470 : initialisation de paramètres importants concernant les possibilités de mouvement, et modifiables, soit pour faciliter le jeu, soit pour le rendre plus ardu. Nous reviendrons sur ces lignes pour étudier l'évolution possible du jeu.

Lignes 520 à 580 : détermination des quatre fenêtres (de 0 à 3) de gestion de l'écran.

Lignes 620 à 670 : préparation des messages sur la fenêtre visualisant l'état des zones.

Lignes 710 à 780 : préparation des messages sur la fenêtre signalant l'état d'avancement du jeu et les paramètres de la météo.

Lignes 870 à 1580 : dessin de l'aéroport vue de dessus.

Lignes 1620 à 1660 : animation d'un avion signalant l'arrivée des premiers avions.

Ligne 1750 : initialisation d'une première variable pour signaler le nombre d'avions à l'arrivée.

Ligne 1760 : il n'y a pas encore d'avion en attente.

Lignes 1770 à 2030 : acquisition et vérification des différents positionnements affectés aux avions qui arrivent.

Ligne 2040 : recherche de la météo pour le prochain tour de jeu.

Ligne 2130 : régénération de la fenêtre visualisant l'aéroport et ses avions en attente.

Ligne 2140 : on recherche la météo.

Ligne 2150 : affichage de l'état d'avancement du jeu et de la météo.

Lignes 2160 à 2260 : attente d'une commande.

Ligne 2310 : validation du choix atterrissage.

Ligne 2320 : validation du choix mouvement.

Ligne 2330 : validation du choix attente.

Ligne 2340 : validation du choix état d'une zone.

Ligne 2350 : validation du choix décollage.

Ligne 2360 : saut au sous-programme de mise à jour des quantités de kérosène dans les réservoirs des avions en attente.

Ligne 2370 : saut au sous-programme de recherche de nouveaux avions en approche.

Lignes 2380 et 2390 : en fonction du choix, on régénère ou non l'écran.

Lignes 2480 à 2690 : si aucun ordre valide n'a été frappé, la variable `flag%` est restée égale à zéro, aussi, on rappelle les ordres possibles.

Lignes 2770 à 2810 : possibilité d'un mouvement préliminaire des avions.

Lignes 2830 à 3050 : acquisition des coordonnées d'un avion à atterrir.

Ligne 3060 : vérification vent nul.

Ligne 3070 : vérification sens du vent.

Ligne 3090 : vérification avion sur cette position.

Lignes 3130 à 3150 : vérification des zones inférieures libres.

Ligne 3190 : libération de la position quittée.

Ligne 3200 : vérification de l'état de la piste.

Lignes 3210 à 3400 : animation de l'atterrissage.

Lignes 3410 à 3420 : mise à jour des paramètres.

Ligne 3430 : vérification si fin du jeu.

Lignes 3500 à 3560 : traitement en cas de piste gelée (une chance sur deux de réussir).

Ligne 3630 à 3650 : erreur pour l'atterrissage : il n'y avait pas d'avion à cette position.

Lignes 3690 à 3780 : gestion d'un atterrissage avec mauvaise direction du vent.

Lignes 3870 à 4240 : séquence de déplacement des avions.

Ligne 4370 : vérification si avion au départ présent.

Ligne 4380 : vérification si la position d'arrivée est libre.

Ligne 4390 : vérification si changement de zone.

Lignes 4400 à 4420 : vérification si le déplacement dans une même zone est possible (non présence d'avion au-dessus ou en dessous selon le sens de déplacement).

Lignes 4430 et 4440 : mise à jour des emplacements.

Lignes 4590 à 4660 : sous-programme d'attente (en cas de conditions météorologiques défavorables).

Lignes 4740 à 4760 : détermination du nombre de nouveaux avions se présentant aux abords de l'aéroport.

Lignes 4780 à 5060 : traitement des positionnements des avions à l'arrivée.

Lignes 5140 à 5380 : affichage de l'état d'une zone.

Lignes 5470 à 5590 : demande et gestion de la confirmation de décollage d'un avion sur la piste Nord.

Ligne 5600 : vérification du sens du vent.

Lignes 5610 à 5740 : calcul de la maîtrise du pilote pour réussir l'atterrissage.

Lignes 5780 à 5860 : vérification et traitement du décollage lorsque la piste est gelée.

Lignes 5900 à 5920 : vérification si la zone Nord est entièrement libre pour un décollage.

Lignes 5930 à 6090 : animation du décollage.

Lignes 6100 à 6120 : mise à jour des variables aériennes, recherche si fin de partie, et retour au programme principal.

Lignes 6220 à 6340 : recherche du type d'avion arrivant ou décollant par génération aléatoire.

Ligne 6380 : détermination aléatoire de la quantité de carburant disponible dans les réservoirs de l'avion qui arrive.

Lignes 6490 à 6540 : détermination aléatoire du sens du vent.

Lignes 6630 à 6670 : détermination aléatoire de l'état de la piste (sèche ou gelée).

Lignes 6750 à 6830 : détermination de la possibilité de collision entre deux avions.

Lignes 6840 à 7020 : animation de la collision.

Lignes 7030 à 7260 : bilan de la collision.

Lignes 7360 à 7660 : régénération de la fenêtre où se trouve le dessin de l'aéroport.

Lignes 7750 à 7900 : mise à jour de la fenêtre d'avancement du jeu.

Lignes 7990 à 8030 : mise à jour des réservoirs des avions par retrait de la valeur 1 à toutes les cases contenant un avion (la quantité de carburant est obtenue en multipliant par 100 le nombre restant).

Lignes 8040 à 8140 : vérification si réservoir vide.

Lignes 8220 et 8230 : interrogation du centre météo de l'aéroport.

Lignes 8310 à 8470 : fin de partie gagnée.

Lignes 8550 à 8760 : animation d'un avion s'écrasant à l'atterrissage.

Lignes 8770 à 9030 : bilan de la catastrophe.

Lignes 9110 à 9440 : animation d'un avion en panne de carburant.

Lignes 9450 à 9750 : bilan de l'accident.

Les principales variables

ATTENTE%(4, 4)

Cette variable tableau représente les zones et couloirs de navigation aérienne entourant les pistes de l'aéroport. Le premier chiffre sera affecté par un nombre codant les zones (1 = Nord, 2 = Ouest, 3 = Sud et 4 = Est). Le deuxième chiffre sera affecté par l'altitude des couloirs aériens de chacune des zones. Dans ce tableau seront rangées les quantités de carburant de chacun des avions s'y trouvant. Si une case est vide, la valeur correspondante dans le tableau sera égale à -1.

NOMBREDECOLLAGE%

Nombre d'avions à faire décoller.

ARRIVEMAXI%

Nombre d'avions maximum pouvant arriver entre chaque tour de jeu.

BOUGEMAXI%

Nombre maximum d'avions que l'on peut déplacer avant une séquence d'atterrissage.

ATTERRIMAXI%	Nombre maximum d'avions que l'ont peut faire atterrir à chaque tour de jeu.
NOMBREARRIVE%	Nombre d'avions qui arrivent au départ du jeu.
NOMBREAVION%	Nombre d'avions qui sont en attente au-dessus des pistes.
NOMBREATTERRIS%	Nombre d'avions ayant atterri.
POSITION%	Variable transitoire mémorisant la zone d'un avion en mouvement.
ALTITUDE%	Variable transitoire mémorisant l'altitude d'un avion en mouvement.
TYPEAVIONS	Variable transitoire recevant le nom d'un avion à l'arrivée ou au décollage.
DIRECTIONVENT%	Codage numérique de la direction du vent (1 = vent nul, 2 = Ouest, 3 = Sud et 4 = Est).
VENTS	Nature du vent (Nul, Ouest, Sud ou Est).
ETATPISTE%	Variable codant l'état de la piste (0 = sèche, 1 = gelée).
FORMEAVION1\$	Dessin d'un avion se dirigeant à gauche.
FORMEAVION2\$	Dessin d'un avion se dirigeant à droite.

Nous passerons sur toutes les variables éphémères utilisées pour les boucles et affichages (I%, J%, **NOMBREBLESSES%**...).

MODIFICATION DES PARAMÈTRES DU JEU

Nous allons ici vous fournir les différentes adresses à modifier, ainsi que les possibilités, pour vous permettre de faire évoluer le jeu, afin de l'adapter au niveau que vous pourrez acquérir au fur et à mesure de votre entraînement.

Le nombre d'avions au décollage

Pour modifier le nombre d'avions au décollage, et ainsi influencer la durée de la partie, il vous faudra vous reporter à la ligne 410 :

410 nombredecollage% = 15

Si vous baissez ce nombre, vous raccourcirez la durée, et *vice versa*. Avec ce nombre, la durée moyenne de jeu varie entre 15 et 30 minutes

selon votre chance, mais nous reconnaissons tout de même que le jeu est difficile.

Dans un premier temps, vous pourrez affecter la valeur 0, pour vous entraîner uniquement à faire atterrir des avions.

Le nombre d'avions à l'arrivée

En modifiant la ligne 430, vous pourrez modifier le nombre maximum d'avions qui peuvent arriver entre chaque tour de jeu.

430 arrivemaxi% = 2

Cette valeur est traitée en ligne 4760 :

4760 local% = INT (RND *arrivemaxi%) + 1

Afin de garder un certain intérêt au jeu, nous vous recommandons de ne pas annuler la valeur de la variable, et d'en laisser au minimum 1.

L'auteur s'est entraîné avec une valeur égale à 4, et vous conseille de ne pas essayer, sous peine de crise de nerfs.

Le nombre de mouvements avant atterrissage

Pour modifier ce paramètre, vous devez affecter la ligne 450 :

450 bougemaxi% = 1

Une valeur de 2 devrait dans un premier temps vous faciliter la tâche.

L'intérêt d'augmenter cette variable baisse rapidement avec l'augmentation du nombre ; tous les mouvements n'étant pas utilisés.

Le nombre d'atterrissages autorisés

La ligne 470 vous permettra de modifier le nombre maximum d'avions pouvant atterrir lors d'une phase d'atterrissage.

470 atterrimaxi% = 2

Le nombre 2 est ici suffisant pour une difficulté correcte. Néanmoins, si vous avez peur de vous retrouver avec un espace aérien saturé d'avions en attente, vous pouvez augmenter ce nombre à 3 ou 4.

Il va sans dire que baisser cette valeur à 1 augmente considérablement la difficulté, si vous avez un nombre supérieur d'avions à l'arrivée, et que positionner cette variable à zéro rend le jeu impossible à gagner.

Le nombre d'avions au départ

Dans notre version du jeu, cinq avions se présentent dès le début du jeu pour être placés en attente. Cette valeur est spécifiée en ligne 1750 :

```
1750 nombrearrive% = 5
```

Vous pouvez à votre aise modifier cette valeur, si vous vous sentez très fort, ou si l'encombrement des voies aériennes vous effraie.

Il va sans dire que la valeur 16 est la limite maximale, car au-delà, il n'y a plus de couloirs aériens libres, et c'est la collision assurée.

L'arrivée de nouveaux avions

Les lignes concernées (4740 et 4750) influencent la chance (ou malchance ?) de voir arriver de nouveaux avions entre chaque tour de jeu :

```
4740 local % = INT (RND * 2)
```

```
4750 IF local% = 0 THEN RETURN
```

local% est une variable transitoire qui va déterminer si des avions se présentent ou non.

Dans notre cas, elle peut prendre la valeur 0 ou 1 car RND fournit un nombre compris entre 0 et 1, ces valeurs extrêmes n'étant pas atteintes. Ainsi vous avez une chance sur deux de voir s'exécuter l'instruction RETURN, et donc de ne pas rencontrer d'avion.

Si vous désirez voir plus souvent des avions à l'arrivée, vous augmenterez le chiffre 2 de la ligne 4740 :

```
4740 local% = INT (RND * nombrechance)
```

Vous aurez ainsi (nombrechance - 1) chance sur nombrechance de devoir trouver de la place aux avions arrivant (ou encore une chance sur nombrechance de ne pas voir de nouveaux avions se présenter).

Si vous fixez cette valeur à 1, jamais vous ne rencontrerez de nouveaux avions... l'intérêt ?

Par contre, si vous voulez limiter la chance de voir de nouveaux avions, il vous faudra modifier les deux lignes 4740 et 4750, comme ci-dessous :

```
4740 local% = INT (RND * nombrechance)
```

```
4750 IF LOCAL% <> 0 THEN RETURN
```

Ainsi, cette fois-ci, plus vous augmenterez la valeur de nombrechance, plus vous diminuerez les possibilités de voir de nouveaux avions.

L'autonomie en carburant

La ligne 6380 détermine pour chaque avion sa quantité de carburant :

6380 carburant% = INT (RND * 8) + 2

Les valeurs désignées sont en fait à multiplier par 100 pour connaître les capacités réelles.

Ainsi, les avions arrivent avec un minimum de 200 litres de kérosène (la valeur 2 impose cette quantité) et peuvent avoir jusque $700 + 200 = 900$ litres (INT (RND * 8) peut atteindre au maximum 7).

Vous pouvez donc modifier les deux paramètres, quantité minimale et quantité maximale en modifiant respectivement les nombres 2 et 8 (attention le nombre 2 augmente aussi la quantité maximale).

Augmenter fortement ces valeurs réduira, certes, l'angoisse de la panne sèche, mais vous incitera à garder plus longtemps vos avions en attente, au profit des décollages, ce qui aura tendance à provoquer un encombrement rapide de l'espace aérien.

Nous vous proposons de modifier cette ligne par :

6380 carburant% = INT (RND * 9) + 4

ce qui vous permettra d'obtenir un confort de jeu intéressant, sans trop prendre de risque d'encombrement.

Le sang-froid des pilotes

Il vous sera possible en suivant ces lignes d'augmenter l'expérience des pilotes quant à leurs réactions face à des situations météorologiques défavorables, lors de l'atterrissage.

- *L'assurance face au vent*

Les lignes 3740 et 3750 gèrent les chances de réussite des manœuvres dans les atterrissages par vent défavorable.

3740 local% = INT (RND * 2) + altitude%

3750 IF local% <> 2 THEN GOTO 8550

Vous vous apercevez que l'altitude de départ de l'avion influence la chance (l'envoi à la ligne 8550 provoque la catastrophe). Ainsi, pour une altitude égale à 1, vous avez une chance sur deux de vous en sortir, pour une altitude de 2, une chance sur 3, pour 3, une chance sur 4 et pour l'altitude 4, une chance sur 5. Déjà pour avoir une chance sur 2 à chaque coup, remplacer la variable `altitude%` par la valeur 1.

Si vous estimez encore ne pas avoir assez de chance, vous pourrez modifier les deux lignes par :

```
3740 local% = INT (RND * nombrechance)
3750 IF LOCAL% = 0 THEN GOTO 8550
```

Ainsi, en remplaçant nombrechance par la valeur 5, vous avez une « malchance » sur cinq pour que le pilote rate sa manœuvre.

N'augmentez tout de même pas trop ce nombre, sinon quel serait l'intérêt de faire intervenir le sens du vent dans la gestion des atterrissages ?

- *L'assurance face au gel*

Les lignes 3540 et 3550 assurent le calcul des chances d'atterrissage au cas où la piste est gelée :

```
3540 local% = INT (RND * 2)
3550 IF local% = 0 THEN GOTO 8550
```

Avec ces valeurs, une chance sur deux d'atterrir correctement sur une piste gelée. Pour augmenter vos chances, augmentez le nombre 2 de la ligne 3540.

Si par contre vous désirez les diminuer, augmentez ce nombre, mais remplacez la ligne 3550 par :

```
3550 IF LOCAL% < > 0 THEN GOTO 8550
```

Modifier la météo

Comme vous pourrez le remarquer, la météo n'est pas toujours clémente, et nous vous proposons ici de modifier la météo afin de choisir vous-mêmes la saison durant laquelle vous allez assurer le poste de contrôleur aérien.

- *Le vent*

La gestion de la direction du vent s'effectue entre les lignes 6420 et 6540, avec pour lignes importantes :

```
6490 directionvent% = INT (RND * 4) + 1
6500 IF directionvent% = 1 then vent$ = "Nul"
6510 IF directionvent% = 2 then vent$ = "Ouest"
6520 IF directionvent% = 3 then vent$ = "Sud"
6530 IF directionvent% = 4 then vent$ = "Est"
```

Si vous désirez, en plein été avoir plus souvent un vent nul, qui vous permet les atterrissages sur n'importe quelle piste, et les décollages, modifiez la ligne 6490 et ajoutez la ligne 6495 :

```
6490 directionvent% = INT (RND * nombrechance) + 1
6495 IF directionvent% > nombrechance then
    directionvent% = 1
```


Vous veillerez à prendre une valeur de nombrechance bien supérieure à 4.

- *Le gel*

Le gel des pistes est géré en lignes 6560 à 6670, et notamment par la ligne 6630 :

6630 etatpiste% = INT (RND * 7)

La ligne 6650 détermine si la piste est gelée en comparant la valeur de `etatpiste%` à 1, ce qui donne 6 chances sur 7 d'avoir une piste correctement praticable.

Si vous désirez travailler au cœur d'un hiver particulièrement rigoureux, modifier la valeur 7 en la diminuant (sa valeur minimale sera 2).

Vous pourrez aussi l'augmenter afin de vous croire en demi-saison, aux approches de l'hiver.

Si par contre, vous prenez votre poste en plein été, le gel est impossible, aussi remplacez la ligne 6630 par :

6630 etatpiste% = 0

Voilà, nous l'espérons, un jeu qui vous amusera longtemps, de par ses rebondissements de situations et ses possibilités d'évolution, et qui complètera votre panoplie de simulateurs de vols, bien que ne fonctionnant pas en temps réel, ce qui aurait rendu le jeu trop difficile.

